

Apellidos:

Nombre:

Notas

1. En la evaluación de los ejercicios se tendrá en cuenta la simplicidad y la eficiencia.
2. Hay que describir las definiciones auxiliares.

Ejercicio 1 [7 puntos] Definir la relación `partición(+L, -L1, -L2)` que se verifique si `L1` y `L2` forman una partición de la lista de números `L` (es decir, los elementos de `L` se distribuyen entre `L1` y `L2` de forma que la suma de los elementos de `L1` es igual a la suma de los elementos de `L2`). Por ejemplo,

```
?- partición([10,5,7,4,2,14],L1,L2).
L1 = [10, 5, 4, 2]    L2 = [7, 14] ;
L1 = [10, 7, 4]      L2 = [5, 2, 14] ;
L1 = [5, 2, 14]      L2 = [10, 7, 4] ;
L1 = [7, 14]         L2 = [10, 5, 4, 2] ;
No
?- partición([1,2,5,7],L1,L2).
No
?- findall(X,between(1,4000,X),L), partición(L,L1,L2).
L1 = [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12 | ...]
L2 = [2828, 2829, 2830, 2831, 2832, 2833 | ...]
```

Solución: La definición de `partición` es

```
partición(L,L1,L2) :-
    suma_lista(L,N),
    N mod 2 == 0,
    M is N/2,
    subconjunto_suma(L1,L,M),
    diferencia(L,L1,L2).
```

La relación `suma_lista(+L, -N)` se verifica si `N` es la suma de los números de la lista `L`. Por ejemplo,

```
?- suma_lista([2,3,5],N).
N = 10
```

La definición de `suma_lista` es

```
suma_lista([],0).
suma_lista([X|L],N) :-
    suma_lista(L,M),
    N is X+M.
```

La relación `subconjunto_suma(-L1, +L2, +M)` se verifica si `L1` es un subconjunto de `L2` de suma `M`. Por ejemplo,

```
?- subconjunto_suma(L,[1,2,3],3).
L = [1, 2] ;
L = [3] ;
No
```

La definición de subconjunto_suma es

```
subconjunto_suma([],[],0).
subconjunto_suma([X|L1],[X|L2],M) :-
    X =< M,
    M1 is M-X,
    subconjunto_suma(L1,L2,M1).
subconjunto_suma(L1,[_|L2],M) :-
    subconjunto_suma(L1,L2,M).
```

La relación diferencia(+L1,+L2,-L3) se verifica si L3 es la diferencia de los conjuntos L1 y L2. Por ejemplo,

```
diferencia([],_,[]).
diferencia([X|L1],L2,L3) :-
    memberchk(X,L2),!,
    diferencia(L1,L2,L3).
diferencia([X|L1],L2,[X|L3]) :-
    diferencia(L1,L2,L3).
```

Ejercicio 2 [3 puntos] Se considera la siguiente gramática de cláusulas definidas

```
s(X) --> n(X).
s(X) --> n(X1), s(X2), {X is X1+X2}.
n(X) --> [X], {number(X)}.
```

1. Determinar las respuestas a las siguientes preguntas

```
?- s(X,[1,7,2],[]).
?- s(X,[1,b,2,c],[]).
```

2. Explicar el sentido de la gramática; es decir, explicar la respuesta a la pregunta

```
?- s(X,L,[]).
```

.....

Solución:

1. La respuesta es $X = 10$.
2. La respuesta es No.
3. Si L es un lista de números, entonces X es la suma de los elementos de L. En caso contrario, la respuesta es No.